
ABSTRACT: We describe a Macaulay2 package for computations in prime characteristic commutative
algebra. This includes those for Frobenius powers and roots, p−e-linear and pe-linear maps, singularities
defined in terms of these maps, different types of test ideals and modules, and ideals compatible with a
given p−e-linear map.

1. INTRODUCTION. This paper describes methods for computing objects and numerical invariants in
prime characteristic commutative algebra, implemented in the package TestIdeals for the computer al-
gebra system [Macaulay2]. A ring R of prime characteristic p comes equipped with the Frobenius
endomorphism

F : R −→ R given by F(x)= x p,

which is the basis for many constructions and methods. Notably, the Frobenius endomorphism can be
used to detect whether a ring is regular [Kunz 1969], and further, to quantify how far a ring is from being
regular, measuring the severity of a singularity.

In this direction, two notable applications of the Frobenius endomorphism are the theory of tight
closure (see [Hochster and Huneke 1990; Hochster 2007] for an introduction) and the resulting theory
of test ideals (see the survey [Schwede and Tucker 2012]). These methods are used by a wide group of
commutative algebraists and algebraic geometers.

The TestIdeals package was started during a Macaulay2 development workshop in 2012, hosted by
Wake Forest University. The package, at that time called PosChar, aimed to provide a unified and efficient
set of tools to study singularities in characteristic p > 0, and in particular, collecting and implementing
several algorithms that had been described in research papers. Since then, PosChar has been split into two

packages, TestIdeals and [FrobeniusThresholds] (see [Hernández et al. 2019]), and much more function-
ality has been added by many contributors, during several more Macaulay2 development workshops.1

Starting at least with Kunz [1969] and Fedder [1983], it has been known that the Frobenius en-
domorphism offers effective methods for measuring singularities in positive characteristic. However,
the algorithms that form the basis of this package are the methods for computing Frobenius roots, ?-
closures, and parameter test ideals. These first appeared in [Katzman 2008; 2010; Blickle et al. 2008;
2009]. Another algorithm in [Katzman and Schwede 2012] for computing prime ideals compatible with
a given p−e-linear map was implemented and used to produce the examples in that paper. The methods
for computing test ideals and test modules that were used implicitly in papers such as [Blickle et al.
2010; Katzman et al. 2009; Schwede and Tucker 2014a] became implementable via the Frobenius roots
functionality. Algorithms for computing F-pure thresholds from [Hernández 2014; 2015; Hernández
and Teixeira 2017] form some of the key methods in the forthcoming FrobeniusThresholds package.

2. FROBENIUS POWERS AND FROBENIUS ROOTS. Let R denote a commutative ring of prime charac-
teristic p.

Definition 2.1. Given an ideal I ⊆ R and an integer e ≥ 0, we define the pe-th Frobenius power of I ,
denoted I [p

e
], to be the ideal generated by the pe-th powers of all elements of I.

It is easy to see that if I is generated by g1, . . . , g`, then I [p
e
] is generated by g pe

1 , . . . , g pe

` .

Definition 2.2. Given an ideal I ⊆ R and an integer e ≥ 0, we define the pe-th Frobenius root of I ,
denoted I [1/pe

], to be the smallest ideal J such that I ⊆ J [p
e
], if such an ideal exists.

Frobenius roots always exist in polynomial and power series rings, and in F-finite regular rings (see
[Blickle et al. 2008, §2] and [Katzman 2008, §5]). Below is an example of a computation of a Frobenius
root in a polynomial ring. In Section 2A we describe the main ideas behind such computations.

i1 : R = ZZ/5[x,y,z];

i2 : I = ideal(x^6*y*z + x^2*y^12*z^3 + x*y*z^18);

o2 : Ideal of R

i3 : frobeniusPower(1/5, I)

2 3
o3 = ideal (x, y , z )

o3 : Ideal of R

2A. The mathematics behind the computation of Frobenius roots. We can also describe Frobenius
roots as follows: In a (sufficiently local) regular ring, we have an identification of R with its canonical
module ωR . On the other hand, the Grothendieck dual to the e-iterated Frobenius map2 R −→ R1/pe

1Development workshops hosted by the University of California, Berkeley (2014, 2017), Boise State University (2015), and
the University of Utah (2016).

2In a reduced ring R, it is often convenient to express the Frobenius endomorphism R −→ R as the inclusion R ↪→ R1/p, so
that the source and target are distinguished.

provides a map

T : ωR1/pe −→ ωR, (2-1)

which, using the identification R ∼= ωR , gives us a map

T : R1/pe
−→ R.

It is not difficult to prove that T (I 1/pe
)= I [1/pe

], where I 1/pe
⊆ R1/pe

is simply the ideal of pe-th roots
of the elements of I. In the case that

R = K[x1, . . . , xd ],

where K is a perfect field, R1/pe
is a free R-module with basis consisting of the monomials

xλ/pe
= xλ1/pe

1 · · · xλd/pe

d ,

where 0≤ λi ≤ pe
− 1. Furthermore, the map T : R1/pe

−→ R is simply the projection defined as follows:

T (xλ1/pe

1 · · · xλd/pe

d )=

{
1 if λi = pe

− 1 for all i,
0 otherwise.

Using this, it is not difficult to see that if

f 1/pe
=

∑
λ

fλxλ/pe
, (2-2)

where λ runs over the tuples (λ1, . . . , λd) with 0≤ λ j ≤ pe
−1, and fλ ∈ R, then ( f )[1/pe

]
= T (( f )1/pe

)

is the ideal generated by the fλ. We can then compute I [1/pe
] for more general I by linearity: if I =

( f1, . . . , fm), then I [1/pe
]
=
∑
( fi )
[1/pe

].

Complexity. The computation of Frobenius roots is the main component behind many of the methods
in TestIdeals. Hence, it is important to understand how this is implemented, and its computational
complexity.

The computation of Frobenius roots of ideals reduces to the case of principal ideals, and its complexity
grows linearly with the number of generators of the ideal. Furthermore the calculation of ( f )[1/pe

]

reduces to finding the terms in the right-hand side of (2-2), which essentially amounts to taking each
term in f , computing the pe-th root of its coefficient, and dividing the monomial exponent vector by pe

with remainder. Hence the complexity of computing ( f )[1/pe
] is proportional to the number of terms

in f , and is independent of its degree. We emphasize that the calculation of Frobenius roots does not
involve the calculation of Gröbner bases.

2B. Dual to Frobenius on quotient rings and Frobenius on top local cohomology. For any reduced
ring R with finite e-iterated Frobenius map identified with R ↪→ R1/pe

, we always have a map

TR : ωR1/pe −→ ωR.

At a maximal ideal m of height d, this map is Matlis dual to the Frobenius map on H d
m(R). Hence we

can study local cohomology by studying this map. If R = S/I, where S is a polynomial ring over a
finite field, we can implement this map [Fedder 1983; Katzman 2008]. We briefly explain the case of a
hypersurface here.

If I = ( f ), again ωR ∼= R = S/I, and the action of TR can be computed on S. In fact, if J ⊆ R ∼= ωR

has preimage J ⊆ S, then setting u = f pe
−1, we have that TR(J

1/pe

) is the image of (u J )[1/pe
] in R.

More generally, for non-hypersurfaces, the analog to u is chosen to be an element that maps onto the
generator of the cyclic R-module

(I [p
e
]
: I )∩ (�[p

e
]
:�)

I [pe]
,

where � is an ideal of S with I ⊆� and whose image in R is isomorphic to the canonical module of R;
see [Katzman 2008] for details. Here we compute two examples.

i1 : S = ZZ/5[x,y,z];

i2 : f = x^3 + y^3 + z^3;

i3 : u = f^(5 - 1);

i4 : frobeniusRoot(1, ideal u)

o4 = ideal (z, y, x)

o4 : Ideal of S

i5 : S = ZZ/7[x,y,z];

i6 : f = x^3 + y^3 + z^3;

i7 : u = f^(7 - 1);

i8 : frobeniusRoot(1, ideal u)

o8 = ideal 1

o8 : Ideal of S

The above example has shown that the Frobenius map on top local cohomology of the cone over the
Fermat elliptic curve is injective in characteristic 7 (the dual map ωR1/p −→ ωR is surjective) and not
injective in characteristic 5 (the dual map ωR1/p −→ ωR is not surjective).

2C. A generalization of Frobenius powers and roots. Let R be an F-finite regular ring. We can extend
the definition of Frobenius powers as follows.

Definition 2.3 [Hernández et al. 2018]. Let I ⊆ R be an ideal.

(a) If n is a positive integer with base p expansion n = d0+ d1 p+ · · ·+ dr pr, we define

I [n] = I d0(I d1)[p] · · · (I dr )[p
r
].

(b) If t is a nonnegative rational number of the form t = a/pe, we define I [t] = (I [a])[1/pe
].

(c) If t is any nonnegative rational number, and {an/pen }n≥1 is a sequence of rational numbers con-
verging to t from above, we define I [t] to be the stable value of the nondecreasing chain of ideals
{I [an/pen ]

}n≥1.

i1 : p = 3;

i2 : R = ZZ/p[x,y];

i3 : m = monomialIdeal(x, y);

o3 : MonomialIdeal of R

i4 : I = m^5;

o4 : MonomialIdeal of R

i5 : t = 3/5 - 1/(5*p^3);

i6 : frobeniusPower(t, I)

2 2
o6 = ideal (y , x*y, x )

o6 : Ideal of R

i7 : frobeniusPower(t - 1/p^5, I)

o7 = ideal (y, x)

o7 : Ideal of R

The generalized Frobenius powers of an ideal are test ideals (see Section 4) of sufficiently general
linear combinations of generators of that ideal. Thus, the ideals computed above are test ideals of generic
quintics in two variables in characteristic p = 3, and these computations suggest that 3/5− 1/(5p3) is
a higher F-jumping exponent of such quintics, which is indeed the case for any p ≡ 3 mod 5 (see
[Hernández et al. 2020]). To illustrate, compare the computations above with the following.

i8 : S = ZZ/3[a..f,x,y];

i9 : G = a*x^5 + b*x^4*y + c*x^3*y^2 + d*x^2*y^3 + e*x*y^4 + f*y^5;

i10 : testIdeal(t, G)

2 2
o10 = ideal (y , x*y, x )

o10 : Ideal of S

i11 : testIdeal(t - 1/p^5, G)

o11 = ideal (y, x)

o11 : Ideal of S

Caveat. The computations of Frobenius powers with exponents whose denominators are not powers of
the characteristic p are often very slow, and in several instances it is more efficient to introduce more
variables and compute the test ideal of a “very general” linear combination of the generators of the ideal,
as above. The code used for computations of Frobenius powers needs to undergo significant optimization.

2D. Frobenius roots of submodules of free modules. Let S be a polynomial ring or power series ring.
Given a submodule M of the free module Sk, there is a smallest submodule N of Sk that contains M, for
which M ⊆ N [p

e
]. Here, N [p

e
] is the submodule of Sk generated by the vectors of N with all coordinates

raised to the pe-th power. (See [Katzman and Zhang 2014].) Extending the previous definitions, we call
this N the pe-th Frobenius root of M, and denote it by M [1/pe

].

Example 2.4. Let R = Z/pZ[a, b, c, d], and consider the ideals m= (a, b, c, d) and

I = (a, b)∩ (a, c)∩ (c, d)∩ (c+ d, a3
+ bd2).

Then R/I is a 3-dimensional non-Cohen–Macaulay ring. Matlis duality applied to H 2
m(R/I ) with its

natural Frobenius map yields a p−1-linear map U on Ext2R(R/I, R).

i1 : R = ZZ/2[a,b,c,d];

i2 : I = intersect(ideal(a,b),ideal(a,c),ideal(c,d),ideal(c+d,a^3+b*d^2));

o2 : Ideal of R

i3 : f = inducedMap(R^1/I, R^1/frobenius(I));

o3 : Matrix

i4 : E2 = Ext^2(f, R^1)

o4 = {-8} | a4+abc2+abcd a2b |
{-10} | a2cd3 a3cd+a3d2+bcd3 |

o4 : Matrix

i5 : target E2

o5 = cokernel {-8} | 0 a2 b2c2 |
{-10} | c2+d2 d4 a4d2 |

2
o5 : R-module, quotient of R

i6 : source E2

o6 = cokernel {-4} | 0 a bc |
{-5} | c+d d2 a2d |

2
o6 : R-module, quotient of R

The Frobenius map on H 2
m(R/I ) is injective if and only if the image of (Image U )[1/p] in Ext2R(R/I, R)

as computed above is the whole of R2.

i7 : U = matrix entries E2;

2 2
o7 : Matrix R <--- R

i8 : A = image matrix entries relations source E2;

i9 : frobeniusRoot(1, image U)

o9 = image {-2} | 1 0 0 |
{0} | 0 d a |

2
o9 : R-module, submodule of R

The calculation above shows that R/I is not F-injective (see Section 3). Moreover, it shows that
H 2
m(R/I )∼= AnnE2 At, where E is the injective hull of Rm/mRm, and the Frobenius map on H 2

m(R/I )
induced from Frobenius on R is given by U t2, where2 is the induced Frobenius on E . The submodule of
nilpotent elements in H 2

m(R/I ) is given by AnnE2 B t, where B is the smallest submodule of R2 containing
Image A+ Image U such that U B ⊆ B[p]. The method ascendModule can be used to calculate B (see
a detailed description of the similar method ascendIdeal in Section 4).

i10 : B = ascendModule(1, A, U)

o10 = image | 0 a bc |
| c+d d2 a2d |

2
o10 : R-module, submodule of R

3. F -SINGULARITIES. The TestIdeals package includes methods for determining if a ring is F-injective,
F-pure, F-rational, or F-regular.

3A. F-injectivity.

Definition 3.1. A local ring (R,m) is called F-injective if the map H i
m(R)→ H i

m(R
1/p) is injective for

all i > 0. An arbitrary ring is called F-injective if its localization at each prime ideal of R is F-injective.

The function isFInjective determines whether the ring R = S/I is F-injective, where S is a poly-
nomial ring.

i1 : R = ZZ/7[x,y,z]/(x^3 + y^3 + z^3);

i2 : isFInjective R

o2 = true

i3 : R = ZZ/5[x,y,z]/(x^3 + y^3 + z^3);

i4 : isFInjective R

o4 = false

Equivalently, a ring is F-injective if the maps on the cohomology of the dualizing complex

h−iω•R1/p −→ h−iω•R

surject for all i . Note that h−iω•R
∼= Extdim S−i (R, S), the latter of which, Macaulay2 readily computes.

The algorithm used by isFInjective works by checking the surjectivity of the dual Frobenius map

Extdim S−i (R1/p, S)−→ Extdim S−i (R, S).

We begin by computing the map R → R1/p using [PushForward]. Next, the algorithm computes
Exti ( , S) applied to the map from the previous step. Then R is F-injective precisely when the cokernel
of Exti ( , S) is trivial for i .

The Frobenius action on top local cohomology (dual to ωR) is usually computed in a different (faster)
way than the other cohomologies, and this is modified by the CanonicalStrategy option. The default
value for this option is Katzman, which instead of using the PushForward package, relies on the fact
that we already know how to compute the Frobenius action on the canonical module, as described in
Section 2B.

The performance of the algorithm can be improved if the ring of interest is nice enough. If the ring
is Cohen–Macaulay, then setting AssumeCM => true (the default value is false) lets the algorithm
check the Frobenius action only on top cohomology (which is typically much faster, as explained above).
When studying a reduced ring, setting AssumeReduced => true (the default value) avoids computing
the bottom local cohomology, and when studying a normal ring, setting AssumeNormal => true (the
default is false) avoids computing the bottom two local cohomologies.

By default the algorithm checks for F-injectivity at all points of Spec R. However, one can choose to
check F-injectivity only at the origin by setting the option AtOrigin to true.

i1 : R = ZZ/7[x,y,z]/((x - 1)^5 + (y + 1)^5 + z^5);

i2 : isFInjective R -- R is not globally F-injective...

o2 = false

i3 : isFInjective(R, AtOrigin => true) -- but is F-injective at the origin

o3 = true

3B. F-regularity.

Definition 3.2 [Hochster and Huneke 1990; Hara and Watanabe 2002]. A ring R is called strongly F-
regular if the (big) test ideal τ(R) is equal to R. Likewise a pair (R, f t) is called strongly F-regular if
τ(R, f t)= R.

The command isFRegular checks whether a ring or pair is strongly F-regular. Below are two
examples, one of which is F-regular and the other is not.

i1 : R = ZZ/5[x,y,z]/(x^2 + y*z);

i2 : isFRegular R

o2 = true

i3 : R = ZZ/7[x,y,z]/(x^3 + y^3 + z^3);

i4 : isFRegular R

o4 = false

We can also check whether a pair (R, f t) is F-regular.

i1 : R = ZZ/5[x,y];

i2 : f = y^2 - x^3;

i3 : isFRegular(1/2, f)

o3 = true

i4 : isFRegular(5/6, f)

o4 = false

i5 : isFRegular(4/5, f)

o5 = false

i6 : isFRegular(4/5 - 1/100000, f)

o6 = true

All of these checks are done by actually computing the test ideal, as described in Section 4.
If the input ring is Q-Gorenstein, then in each of the cases above, the output is a boolean indicating

if the ring is strongly F-regular. If the input ring is not Q-Gorenstein, then the algorithm can be used
to determine if a ring is strongly F-regular, but cannot prove that a ring is not strongly F-regular (this
latter functionality can, however, be enabled by setting QGorensteinIndex => infinity).

In the case that R is Q-Gorenstein, the algorithm works by computing the test ideal τ of the ring (or
the pair) using testIdeal and checking whether τ = R. In the non-Q-Gorenstein case, the algorithm
checks for strong F-regularity by computing better and better approximations of the test ideal, and
checking whether any of these is the unit ideal. To compute approximations of the test ideal, the algorithm
computes a test element c with testElement and then uses frobeniusRoot to compute the e-th root

of c(I [p
e
]
: I ); appropriate modifications are made for pairs. If at any step the approximation is the unit

ideal, then the algorithm returns true. Otherwise the algorithm continues checking for each e until a
specified limit is reached. The default limit is 2, and can be changed using the option DepthOfSearch.

A number of options can be used to speed up the performance of some of the internal functions. The
option AssumeDomain can be used if R is an integral domain, FrobeniusRootStrategy chooses a
strategy for internal frobeniusRoot calls, MaxCartierIndex sets the maximum Gorenstein index to
search for when working with a Q-Gorenstein ambient ring, and QGorensteinIndex allows the user to
specify the Q-Gorenstein index of the ring.

The default behavior of isFRegular is that it checks for strong F-regularity globally. If the option
AtOrigin is set to true, the algorithm will only check F-regularity at the origin, by checking whether
the computed test ideal is in the irrelevant ideal. Below are examples for both a ring and a pair.

i1 : R = ZZ/7[x,y,z]/((x - 1)^3 + (y + 1)^3 + z^3);

i2 : isFRegular R -- R is not globally F-regular...

o2 = false

i3 : isFRegular(R, AtOrigin => true) -- but is F-regular at the origin

o3 = true

i4 : R = ZZ/13[x,y];

i5 : f = (y - 2)^2 - (x - 3)^3;

i6 : isFRegular(5/6, f) -- (R,f^(5/6)) is not F-regular...

o6 = false

i7 : isFRegular(5/6, f, AtOrigin => true) -- but is F-regular at the origin

o7 = true

3C. F-purity.

Definition 3.3. A ring R is called F-pure if the inclusion R ↪→ R1/pe
is a pure map, i.e., the tensor of this

map with any R-module remains injective. If R1/p is a finite R-module, this is equivalent to requiring
that the inclusion R ↪→ R1/p split as a map of R-modules.

The function isFPure checks whether a ring is F-pure. Either a ring or a defining ideal can be input,
as seen in the following example.

i1 : R = ZZ/5[x,y,z]/(x^2 + y*z);

i2 : isFPure R

o2 = true

i3 : R = ZZ/7[x,y,z]/(x^3 + y^3 + z^3);

i4 : isFPure R

o4 = true

i5 : S = ZZ/2[x,y,z];

i5 : isFPure ideal(y^2 - x^3)

o5 = false

i6 : isFPure ideal(z^2 - x*y*z + x*y^2 + x^2*y)

o6 = true

The algorithm works by applying Fedder’s criterion [1983], which states that a local ring (R,m) is
F-pure if and only if (I [p] : I ) 6⊆ m[p]. When AtOrigin is set to true, the algorithm checks F-purity
only at the origin, by explicitly checking the above containment. When AtOrigin is set to false, which
is its default value, the algorithm computes the non F-pure locus, by applying frobeniusRoot to I [p] : I.
If the non F-pure locus is the whole ring, the algorithm returns true.

3D. F-rationality.

Definition 3.4. Suppose that R is a Cohen–Macaulay ring and that TR : ωR1/p → ωR is the canonical
dual to the Frobenius. We say that R has F-rational singularities, or simply that R is F-rational, if there
are no nonzero proper submodules M of ωR such that TR(M1/p)⊆ M.

The command isFRational checks if a ring is F-rational.

i1 : S = ZZ/3[a,b,c,d,t];

i2 : M = matrix{{ a^2 + t^4, b, d }, { c, a^2, b^3 - d }};

2 3
o2 : Matrix S <--- S

i3 : I = minors(2, M);

o3 : Ideal of S

i4 : R = S/I;

i5 : isFRational R

o5 = true

The algorithm used by isFRational first checks if the ring is Cohen–Macaulay, unless the option
AssumeCM is set to true. If the ring is not Cohen–Macaulay then false is returned. Next, the algorithm
computes the test module M ⊆ ωR and checks to see if ωR ⊆ M ; see the next section for the description
of a test module.

The options AssumeDomain and FrobeniusRootStrategy can be used to improve the speed of the
testModule computation. By default, these options are set to false and Substitution, respectively.
Finally, if AtOrigin is set to true, then F-rationality is checked only at the origin.

4. TEST IDEALS. In this section, we explain how to compute parameter test modules, parameter test
ideals, test ideals, and HSLG modules.3

4A. Parameter test modules. Given a reduced ring R of finite type over a perfect field k, the Frobenius
map R ↪→ R1/pe

is dual to
T : ωR1/pe −→ ωR.

As in Section 2B, we can represent the canonical module ωR ⊆ R as an ideal, we can write R = S/I, and
so we can find an element u ∈ S1/pe

representing the map T : ωR1/pe −→ ωR; see Section 2B or [Katzman
2008].

3HSLG modules can be used to give a scheme structure to the F-injective or F-pure locus.

Definition 4.1. The parameter test submodule is the smallest submodule M ⊆ ωR (and hence ideal of R,
since M ⊆ ωR ⊆ R) that agrees generically with ωR and that satisfies

T (M1/pe
)⊆ M

for some e > 0 (or equivalently for all e > 0).

Using Macaulay2, we can compute this using the testModule command as follows.

i1 : R = ZZ/5[x,y,z]/(x^4 + y^4 + z^4);

i2 : N = testModule R;

i3 : N#0

2 2 2
o3 = ideal (z , y*z, x*z, y , x*y, x )

o3 : Ideal of R

i4 : N#1

o4 = ideal 1

o4 : Ideal of R

The output of testModule is a sequence, consisting of three items:

(1) The test module itself, given as an ideal of R.

(2) The canonical module that contains the test module, given as an ideal of R. (Note the representation
of the canonical module as an ideal is not unique; it is only unique up to isomorphism, hence it is
important to return this module as well).

(3) The element u described above (not displayed above, since it takes a lot of space).

Note since this ring is Gorenstein, the canonical module is simply represented as the unit ideal. Here
is another example, where the ring is not Gorenstein.

i1 : R = ZZ/5[x,y,z]/(y*z, x*z, x*y);

i2 : N = testModule R;

i3 : N#0

2 2 2
o3 = ideal (z , y , x )

o3 : Ideal of R

i4 : N#1

o4 = ideal (y + 2z, x - z)

o4 : Ideal of R

We briefly explain how this is computed: First, we find a test element.

Remark 4.2 (computation of test elements). We recall that, roughly, an element of the Jacobian ideal
that is not contained in any minimal prime is a test element [Hochster 2007]. We compute test elements
by computing random partial derivatives (and linear combinations thereof) until we find an element
that does not vanish at all the minimal primes. This method is much faster than computing the entire

Jacobian ideal. If it is known that the ring is a domain, setting AssumeDomain => true can speed this
up further.

After the test element c has been identified, we pull back the ideal ωR to an ideal J ⊆ S. Next, we
compute the following ascending sequence of ideals where u represents T : ωR1/pe −→ ωR as above:

J0 := cJ

J1 := J0+ (u J0)
[1/p]

J2 := J1+ (u J1)
[1/p]
= J0+ (u J0)

[1/p]
+ (u1+p J0)

[1/p2
]

J3 := J2+ (u J2)
[1/p]
= J0+ (u J0)

[1/p]
+ (u1+p J0)

[1/p2
]
+ (u1+p+p2

J0)
[1/p3

]

...

(4-1)

As soon as this ascending sequence of ideals stabilizes, we are done. In fact, because this strategy is used
in several contexts, the user can call it directly for a chosen ideal J and u with the function ascendIdeal
(this is done for test ideals below).

We can also compute parameter test modules of pairs (ωR, f t) with t ∈Q≥0. This is done by modifying
the element u when the denominator of t is not divisible by p. When t has p in its denominator, we rely
on the fact (see [Blickle et al. 2008; Schwede and Tucker 2014a]) that

T (τ (ωR, f a))= τ(ωR, f a/p),

frobeniusRoot(1, u ∗ I1)= I2,

where the second line roughly explains how this is accomplished internally. Here I1 is τ(ωR, f a) pulled
back to S and, likewise, I2 defines τ(ωR, f a/p) modulo the defining ideal of R.

Remark 4.3 (optimizations in ascendIdeal and other testModule computations). Throughout the
computations described above, we very frequently use the following fact:

( f p
· J )[1/p]

= f · (J [1/p]).

To access this enhancement, one should try to pass functions like ascendIdeal and frobeniusRoot the
elements and their exponents (see the documentation). In particular, when computing the pe-th Frobenius
root of an ideal of the form f n

· J, one should not multiply out f n and J, nor even raise f to the n-th
power, but rather call frobeniusRoot(e, n, f, J).

4B. Parameter test ideals. The parameter test ideal is simply the annihilator of ωR/τ(ωR). In other
words, it is

(τ (ωR) : ωR).

This can also be described as ⋂
I

(I ∗ : I ),

where I varies over ideals defined by a partial system of parameters, and I ∗ denotes its tight closure.
This latter description is not computable, however.



The TestIdeals package for Macaulay2 101

Example 4.4. In this example we repeat the calculation in [Katzman 2008, §9].

i1 : R = ZZ/2[a..e];

i2 : E = matrix {{a, b, b, e}, {d, d, c, a}};

2 4
o2 : Matrix R <--- R

i3 : I = minors(2, E);

o3 : Ideal of R

i4 : S = R/I;

i5 : J = parameterTestIdeal S

o5 = ideal (c + d, b, a)

o5 : Ideal of S

i6 : J = substitute(J, R);

o6 : Ideal of R

i7 : mingens(J + I)

o7 = | c+d b a de |

1 4
o7 : Matrix R <--- R

4C. Test ideals. For an F-finite reduced ring R = S/I, where S is a regular ring, the (big) test ideal4

of R is the smallest ideal J in R, not contained in any minimal prime, such that for all e > 0 and all
φ ∈ HomR(R1/pe

, R), we have
φ(J 1/pe

)⊆ J.

In the case that R is Gorenstein, HomR(R1/pe
, R) is a cyclic R1/pe

-module generated by 8e, which
corresponds with the map T above based on the identification ωR ∼= R [Blickle and Schwede 2013]. More
generally, if R is Q-Gorenstein with index not divisible by p, then for at least sufficiently divisible e > 0,
such a generating 8e still exists.

If such a 8e exists, it can be identified to a generator of the module

(I [p
e
]
: I )/I [p

e
]

by Fedder’s lemma [1983]. Hence we can find a corresponding5 u ∈ I [p
e
]
: I. In this case, if c ∈ S is the

preimage of a test element of R, then setting I0 = cR, it follows that

τ(R)= ascendIdeal(e, u, I0),

where ascendIdeal is the method explained above, in (4-1). The e here means all Frobenius roots are
taken as multiples of e. In other words, we first compute I0+ (u · I0)

1/p[e]. Then we compute

I0+ (u · I0)
1/p[e]
+ (upe

+1
· I0)

1/p[2e]

etc.
4The notion of test ideals was originally introduced in [Hochster and Huneke 1990] in the context of tight closure in finitely

generated modules, whereas our notion of test ideals arises from tight closure in (“big”) modules which are possibly not finitely
generated. Confusingly, big test ideals are included in (small) test ideals.

5This is done via the function QGorensteinGenerator.

Here is an example (a Z/3Z-quotient, where 3 | (7− 1)), where exactly this logic occurs.

i1 : T = ZZ/7[x,y];

i2 : S = ZZ/7[a,b,c,d];

i3 : f = map(T, S, {x^3, x^2*y, x*y^2, y^3});

i4 : I = ker f;

i5 : R = S/I;

i6 : testIdeal R

o6 = ideal 1

o6 : Ideal of R

However, the term u constructed above can be quite complicated if e> 1 (which happens exactly when
(p− 1)K R is not Cartier). For instance, even in the above example we have an extremely complex u:

i7 : toString QGorensteinGenerator(1, R)

o7 = a^2*b^6*c^12+a^3*b^4*c^13+a^3*b^5*c^11*d+a^4*b^3*c^12*d+a^5*b*c^13*d+b^

12*c^6*d^2+a^3*b^6*c^9*d^2+a^4*b^4*c^10*d^2+a^5*b^2*c^11*d^2+a^6*c^12*d

^2+b^13*c^4*d^3+a*b^11*c^5*d^3+a^2*b^9*c^6*d^3+a^4*b^5*c^8*d^3+a^5*b^3*

c^9*d^3+a^6*b*c^10*d^3+a*b^12*c^3*d^4+a^2*b^10*c^4*d^4+a^3*b^8*c^5*d^4+

a^4*b^6*c^6*d^4+a^5*b^4*c^7*d^4+a^6*b^2*c^8*d^4+a^7*c^9*d^4+a*b^13*c*d^

5+a^2*b^11*c^2*d^5+a^3*b^9*c^3*d^5+a^4*b^7*c^4*d^5+a^5*b^5*c^5*d^5+a^6*

b^3*c^6*d^5+a^7*b*c^7*d^5+a^2*b^12*d^6+a^3*b^10*c*d^6+a^4*b^8*c^2*d^6+a

^5*b^6*c^3*d^6+a^6*b^4*c^4*d^6+a^7*b^2*c^5*d^6+a^8*c^6*d^6+a^4*b^9*d^7+

a^5*b^7*c*d^7+a^6*b^5*c^2*d^7+a^7*b^3*c^3*d^7+a^8*b*c^4*d^7+a^6*b^6*d^8

+a^7*b^4*c*d^8+a^8*b^2*c^2*d^8+a^9*c^3*d^8+a^8*b^3*d^9+a^9*b*c*d^9+a^10*d^10

Therefore, we use a different strategy if either (p− 1)K R is not Cartier or, more generally, if R is
Q-Gorenstein of index divisible by p. In these situations, this alternate strategy typically appears to be
faster. We rely on the observation (see [Blickle et al. 2015]) that

τ(ωR, K R)∼= τ(R).

In fact, by embedding ωR ⊆ R, we can compute g so that τ(ωR, K R) = g · τ(R). We can therefore
find τ(R) if we can find τ(ωR, K R). Next, if K R is Q-Cartier with nK R = divR( f ) for some f ∈ R
and n > 0, then

τ(ωR, K R)= τ(ωR, f 1/n).

Thus we directly compute τ(ωR, f 1/n) via the command testModule(1/n, f). Consider the following
example, a µ3-quotient, which uses the logic described above.

i1 : T = ZZ/3[x,y];

i2 : S = ZZ/3[a,b,c,d];

i3 : f = map(T, S, {x^3, x^2*y, x*y^2, y^3});

o3 : RingMap T <--- S

i4 : I = ker f;

o4 : Ideal of S

i5 : R = S/I;

i6 : testIdeal R

o7 = ideal 1

o7 : Ideal of R

Remark 4.5 (nongraded caveats). It frequently happens that (I [p
e
]
: I )/I [p

e
] is principal but Macaulay2

cannot identify the principal generator (since Macaulay2 cannot always find minimal generators of non-
graded ideals or modules). The same thing can happen when computing the element u corresponding to
the map T : ωR1/p −→ ωR , as described in Section 2B. In such situations, instead of a single u, Macaulay2
will produce u1, . . . , un (all multiples of u, and u is a linear combination of the ui ). Instead of computing
the ideal

(u · J )[1/p],

we compute
(u1 · J )[1/p]

+ · · ·+ (un · J )[1/p],

which will produce the same answer.

We can similarly use the testIdeal command to compute test ideals of pairs, τ(R, f t), and even
mixed test ideals, τ(R, f t1

1 · · · f tn
n ).

4D. HSLG module; computing F-pure submodules of rank-1 Cartier modules. Again, take the maps
T e
: ωR1/pe −→ ωR we have considered throughout this section. From [Hartshorne and Speiser 1977;

Lyubeznik 1997; Gabber 2004], we have the theorem that the descending images

ωR ⊇ T (ωR1/p)⊇ · · · ⊇ T e(ωR1/pe )⊇ T e+1(ωR1/pe+1 )⊇ · · · (4-2)

stabilize for e� 0. The function FPureModule computes the stable submodule in this chain, called the
HSLG module, and returns a sequence containing the following items:

(1) The HSLG module.

(2) The canonical module, embedded (nonuniquely) as an ideal.

(3) The element u representing the map on the canonical module (see Section 2B).

(4) The value of e > 0 at which the descending images in (4-2) stabilize. This is sometimes also called
the HSLG number of the canonical module as a Cartier module.

i1 : R = ZZ/3[x,y,z]/(x^3 + y^4 + z^5);

i2 : L = FPureModule R;

i3 : L#0

2 2 3
o3 = ideal (y*z, x*z, y , x*y, x , z )

o3 : Ideal of R

i4 : L#1

o4 = ideal 1

o4 : Ideal of R

i5 : L#3

o5 = 1

i6 : R = ZZ/3[x,y,z]/(x^3 + y^4 + z^5);

i7 : L = FPureModule R;

i8 : L#0

2 2 3
o8 = ideal (y*z, x*z, y , x*y, x , z )

o8 : Ideal of R

i9 : L#1

o9 = ideal 1

o9 : Ideal of R

i10 : L#3

o10 = 1

More generally, for any ideal J with a map φ : J 1/pe
−→ J, we have that the images

J ⊇ φ(J 1/pe
)⊇ φ2(J 1/p2e

)⊇ · · ·

stabilize as well (in fact, the analogous result even holds for modules [Gabber 2004]).

4E. Computing the level of a polynomial. Another interesting invariant that can be calculated using
this package is the so-called level of a polynomial; more precisely:

Definition 4.6. Let K be an F–finite field of prime characteristic p, and let f ∈ R = K [x1, . . . , xd ]. We
define the level of f as the smallest possible integer e where the descending chain

R = ( f p0
−1)[1/p0

]
⊇ ( f p−1)[1/p]

⊇ ( f p2
−1)[1/p2

]
⊇ · · · ⊇ ( f pi

−1)[1/pi
]
⊇ · · ·

stabilizes.

This invariant was introduced in [Alvarez-Montaner et al. 2005]. It is also essentially the same data as
the HSLG number of the pair (R, f 1) as computed in (4) above (it is that number +1). One interesting
particular case is when K = Fp and f is the defining homogeneous polynomial of a hyperelliptic curve
of genus g; when g = 1, it was proved in [Boix et al. 2015] that the corresponding elliptic curve defined
by f is ordinary if and only if its level is 1, (equivalently, if and only if (R/( f ))(x,y,z) is F-pure, ) and
supersingular if and only if its level is 2. When the genus is at least 2, level 2 is a necessary (but not
sufficient) condition for the curve for being ordinary [Blanco-Chacón et al. 2018]; in this case, one also
has that, if the curve is supersingular, then its level has to be at least 3, so the level can always distinguish
these two properties in any genus. We illustrate these results by means of the following examples.

i1 : R = ZZ/2[x,y,z];

i2 : f = x^3 + y^2*z + y*z^2;

i3 : frobeniusPower(1/2, ideal f)

o3 = ideal (z, y, x)

o3 : Ideal of R

i4 : u = f^3;

i5 : frobeniusPower(1/4, ideal u)

o5 = ideal (z, y, x)

o5 : Ideal of R

i6 : ((FPureModule(1, f))#3) + 1

o6 = 2

This shows that this elliptic curve has level 2; hence it is supersingular. The next example shows a
nonordinary hyperelliptic curve of genus 2 with level 2.

i1 : R = ZZ/11[x,y,z];

i2 : f = y^2*z^3 - x^5 - 2*z^5;

i3 : frobeniusPower(1/11, ideal f^10)

2 3
o3 = ideal (z , x*z, x )

o3 : Ideal of R

i4 : frobeniusPower(1/121, ideal f^120)

2 3
o4 = ideal (z , x*z, x )

o4 : Ideal of R

i5 : ((FPureModule(1, f))#3) + 1

o5 = 2

5. IDEALS COMPATIBLE WITH A GIVEN p−e-LINEAR MAP. Throughout this section, let R denote a
polynomial K[x1, . . . , xn]. In this section we address the following question: given an R-linear map
φ : R1/pe

→ R, what are the ideals I ⊆ R such that φ(I 1/pe
)⊆ I ? We call these ideals φ-compatible.

Recall that HomR(R1/pe
, R) is a principal R1/pe

-module generated by a trace map T ∈HomR(R1/pe
, R),

constructed as in Section 2A (see [Fedder 1983, Lemma 1.6] and [Brion and Kumar 2005, Exam-
ple 1.3.1]).

We can now write our given φ as multiplication by some u1/pe
followed by T and it is not hard to see

that an ideal I ⊆ R is φ-compatible if and only if uI ⊆ I [p
e
].

Theorem 5.1. If φ is surjective, there are finitely many φ-compatible ideals, consisting of all possible
intersections of φ-compatible prime ideals (see [Kumar and Mehta 2009, Schwede 2009, Sharp 2007,
Enescu and Hochster 2008]). In general, there are finitely many φ-compatible prime ideals not containing
√

Imageφ (see [Katzman and Schwede 2012]).

The method compatibleIdeals produces the finite set of φ-compatible prime ideals in the second
claim of Theorem 5.1.

i1 : R = ZZ/3[u,v];

i2 : u = u^2*v^2;

i3 : compatibleIdeals u

o3 = {ideal v, ideal (v, u), ideal u}

o3 : List

The defining condition uI ⊆ I [p
e
] for φ-compatible ideals allows us to think of these in a dual form:

write m= (x1, . . . , xn), E = ERm(Rm/m)= H n
m(R), and let 2 : E→ E be the pe-linear map6 induced

from the Frobenius map on R. If ψ = u2, then ψ AnnE I ⊆ AnnE I if and only if uI ⊆ I [p
e
]. Thus

finding all R-submodules of E compatible with ψ = u2 also amounts to finding all φ-compatible ideals,
where φ = T ◦ u1/pe

.

Example 5.2. We return to Example 4.4. In [Katzman 2008, §9] it is shown that there is a surjection
AnnE I → H 2

m(R/I ) which is compatible with the induced p1-linear map on H 2
m(R/I ) and the p1-linear

map u2 on AnnE I, where u is computed as follows.

i1 : R = ZZ/2[a..f];

i2 : E = matrix {{a, b, b, e}, {d, d, c, a}};

2 4
o2 : Matrix R <--- R

i3 : I = minors(2, E);

o3 : Ideal of R

i4 : S = R/I;

i5 : isCohenMacaulay S

o5 = true

In [Katzman 2008] it is shown that since R/I is Cohen–Macaulay, u can be taken as the generator of
the cyclic module (I [p] : I )∩ (�[p] :�)/I [p] where �⊆ R is an ideal whose image in R/I is a canonical
module of that ring.

i6 : omega = canonicalIdeal S

o6 = ideal (e, d, a)

o6 : Ideal of S

i7 : omega = substitute(omega, R) + I;

o7 : Ideal of R

i8 : u = intersect((frobenius I):I, (frobenius omega) : omega);

o8 : Ideal of R

i9 : u = compress((gens u) % gens(frobenius I))

o9 = | a3bc+a3bd+a2cde+abcde+abd2e+b2d2e+cd2e2+d3e2 |

1 1
o9 : Matrix R <--- R

i10 : u = first first entries u

3 3 2 2 2 2 2 2 3 2
o10 = a b*c + a b*d + a c*d*e + a*b*c*d*e + a*b*d e + b d e + c*d e + d e

o10 : R

Now we can compute all annihilators of R-submodules of E stable under the p1-linear map uT.

i11 : L = compatibleIdeals u;

i12 : print \ L;

3 3 2 2 2 2 2 2 3 2
ideal(a b*c + a b*d + a c*d*e + a*b*c*d*e + a*b*d e + b d e + c*d e + d e )

6That is, 2 is additivive and 2(ra)= r pe
2(a) for all a ∈ E and r ∈ R.

2
ideal (a + b, a + d*e)

ideal (e, a, d)

ideal (e, d, a, c)

ideal (e, d, c, b, a)

ideal (e, d, b, a)

ideal (e, b, a)

ideal (e, b, a, c + d)

2
ideal (c + d, a + b, b + d*e)

ideal (d, a)

ideal (d, a, c)

ideal (d, c, b, a)

ideal (d, b, a)

We can also compute all annihilators of R-submodules of H 2
m(R/I ) stable under the p1-linear map

induced from the Frobenius map on R/I.

i13 : print \ unique apply(L, J -> J : omega);

3 3 2 2 2 2 2 2 3 2
ideal(a b*c + a b*d + a c*d*e + a*b*c*d*e + a*b*d e + b d e + c*d e + d e )

2
ideal (a + b, b + d*e)

ideal (e, d, a)

ideal 1

ideal (e, b, a)

ideal (e, c + d, b, a)

2
ideal (c + d, a + b, b + d*e)

ideal (d, a)

ideal (d, c, a)

ideal (d, c, b, a)

ideal (d, b, a)

6. FUTURE PLANS. In [Katzman and Zhang 2014], the algorithms behind the methods in Section 5
were extended to compute prime annihilators of submodules of Artinian modules compatible with a
given pe-linear map. This would require, among other things, a faster implementation of our method for
finding Frobenius roots of submodules of free modules.

On the other hand, it should be possible to compute test ideals of pairs (R, at) even when a is not prin-
cipal. One strategy to do this is outlined in [Schwede and Tucker 2014b] although certain improvements
can be made.

We hope to achieve all these things during a future Macaulay2 workshop. We also want to bring to the
reader’s attention the package FrobeniusThresholds, which computes F-pure thresholds, F-thresholds,
F-jumping numbers and more!



108 Boix, Hernández, Kadyrsizova, Katzman, Malec, Robinson, Schwede, Smolkin, Teixeira and Witt

ACKNOWLEDGEMENTS. The authors thank other contributors of the TestIdeals package, namely Erin
Bela, Juliette Bruce, Drew Ellingson, Matthew Mastroeni, and Maral Mostafazadehfard. We also thank
the referees for numerous useful comments on the package and on this paper, and Thomas Polstra for
pointing out typos in a previous version of the paper. This paper, and the finishing touches to the
TestIdeals package, were made at the University of Utah in 2018, and the visiting authors thank the
Department of Mathematics for its hospitality. The authors also thank the referees for their numerous
useful comments on both this paper and the code. Additional work on the package was done at the
Institute of Mathematics and its Applications, and the authors are grateful for its generous support for a
2019 Coding Sprint on the FrobeniusThresholds package.

SUPPLEMENT. The online supplement contains version 1.01 of the TestIdeals package.

REFERENCES.
[Alvarez-Montaner et al. 2005] J. Alvarez-Montaner, M. Blickle, and G. Lyubeznik, “Generators of D-modules in positive
characteristic”, Math. Res. Lett. 12:4 (2005), 459–473. MR Zbl

[Blanco-Chacón et al. 2018] I. Blanco-Chacón, A. F. Boix, S. Fordham, and E. S. Yilmaz, “Differential operators and hyperel-
liptic curves over finite fields”, Finite Fields Appl. 51 (2018), 351–370. MR Zbl

[Blickle and Schwede 2013] M. Blickle and K. Schwede, “p−1-linear maps in algebra and geometry”, pp. 123–205 in Com-
mutative algebra, edited by I. Peeva, Springer, 2013. MR Zbl

